


Application Note		資料作成 140128	資料番号 an2k-009
テーマ	MPG-2314 のエンコーダ入力と制御例		
使用機器	MPG-2000 シリーズ CPU、MPG-2314、2 相ロータリエンコーダ、ステップモータ		

■機器構成



■概要

- MPG-2314 は 4 軸のエンコーダ入力を備えています(X,Y 軸は J6 コネクタ入力、U,Z は J7 コネクタ入力 でオプションケーブル DF13-C8 を使用)。
- 2 相または Up/Down をコマンドで選択可能です。

■コマンド例

PG n	MPG-2314 アサイン。エンコーダ入力をする場合にも必要です。
INSET X_A PHASE1	2 相エンコーダ入力 1 通倍
INSET X_A UP_DWN	Up/Down 入りに切り替える (デフォルトは 2 相)
INSET X_A CMP_CNT	COMP レジスタとカウント値を比較する
CLRPOS -1	X, Y, U, Z カウンタクリア
STPS X_C 0	X カウンタを 0 に設定する
STPS X_C 1000	X カウンタを 1000 に設定する
X(-1)	X カウンタの値を返す
X(-2, 1)	X カウンタの値を返してカウンタをクリアする
CMP_C(X_A)	COMP レジスタと X カウンタの比較結果を返す

■サンプル1 ロータリエンコーダ カウンタの設定と読み込み

X(-1),Y(-1)(U(-1),Z(-1))で現在のカウンタ値が読めます。
回転軸やリニアガイドの位置の検出に応用できます。

```

PG 0
INSET PHASE1           /* 1 通倍 Resolution:1000P/R のエンコーダなら 1 回転で 1000 カウント
/* INSET PHASE2       /* 2 通倍           "           2000 カウント
/* INSET PHASE4       /* 4 通倍           "           4000 カウント
CLRPOS -1             /* カンタクリア
OUT 0 0
DO
  NOW_XC=X(-1)        /* カウンタ値読み込み
  IF (NOW_XC%100)==0 THEN
    OUT @SW(0) 0      /* 出力反転 on/off
    PRINT NOW_XC SW(0) /* 表示
    WAIT NOW_XC<>X(-1)
  END_IF
LOOP

```

・ 実行結果

ロータリエンコーダを回すと 100 カウント毎に出力 0 を on/off します

```

0 1
100 0
200 1
200 0

```

```

100 1
0 0
-100 1
-200 0

```

■サンプル2 ロータリエンコーダでステップモータを回す

RMVS に X(-2,1) を与えると、その時のカウンタ量だけパルスを発生し、その後カウンタをクリアします。これを繰り返せばステップモータがロータリエンコーダに追従して回転します。ティーチングや位置出しに応用できます。
 ※ACCEL 設定、エンコーダの回転速度等の条件により誤差が出ることがあります。

```

PG 0
INSET X_A PHASE1          /* 1 通倍
ACCEL ALL_A 10000 100 10 /* leng, lo_pps は小さい方が追従がよいみたい
CLRPOS -1                 /* カンタクリア
CLRPOS                    /* モータの現在位置クリア

DO
  RMVS X(-2,1) 0 0 0      /* パルス発生してカウンタクリア
  WAIT RR(X_A)==0         /* モータの現在位置
  CXP=X(0)
  IF (CXP%1000)==0 THEN
    ON 0
  ELSE
    OFF 0
  END_IF
  SWAP
LOOP

```


・ 実行結果

ロータリエンコーダを回すと 1000 の倍数時に出力 0 がオンします。

■サンプル3 動作エリア設定と検出

予め設定した動作エリア内にいるのか否かを検出します。動作エリア内なら CMP_C() は 0 を返し、動作エリアを越えると CMP_C() は 1 または 2 を返します。

CMP_C の bit	1	0
状態	X(-1) < COMPm : 1 X(-1) >= COMPm : 0	X(-1) >= COMPp : 1 X(-1) < COMPp : 0



※CMP_C() は MCX314 の RR1 レジスタの D1 (CMP-) と D0 (CMP+) の状態を返す

```

PG 0
INSET X_A CMP_CNT|PHASE1 /* RANGE 設定値とエンコーダカウンタを比較する
ACCEL X_A 5000
STPS X_C 0                /* エンコーダカウンタクリア
COMPp=1000                /* RANGE 上限 (COMP+レジスタ)
COMPm=-1000               /* RANGE 下限 (COMP-レジスタ)
RANGE X_A COMPp COMPm    /* エリア設定

/* STPS X_A 0              /* ※モータを回す場合=現在位置クリア
/* RMVC X_A CW             /* ※モータを回す場合=CW回転
WAIT CMP_C(X_A) <> 0      /* 比較結果(0または1または2)
/* STOP X_A STP_I         /* ※モータを回す場合=停止
SELECT_CASE CMP_C(X_A)
CASE 1 : PRINT "1=Upper than COMPp" X(-1)
CASE 2 : PRINT "2=Lower than COMPm" X(-1)
CASE_ELSE : PRINT "?"
END_SELECT
END

```

・ 実行結果

ロータリエンコーダを

CW 方向に回してエリアを越えると
 1=Upper than COMPp 1000


CCW 方向に回してエリアを越えると
2=Lower than COMPm -1001

■サンプル4 移動中にトリガ信号を出す。

通常、カウンタを読んでI/Oを制御するにはサンプル1の方法で可能と思われませんが、回転が速い場合や、精度を要求される場合はINTA_ON,INTA_OFF,INTB_ON,INTB_OFFを試してください。

これらの目的は移動中にトリガ信号を出すことです。

MCX314の「論理/実位置カウンタとCOMPレジスタの比較」で発生した割り込み信号をメインCPUが1mSec毎に監視して指定したポートをON/OFFします。



```

INTA_ON VOID /* 一旦割り込み無効化
PG 0
INSET X_A CMP_CNT|PHASE1 /* RANGE 設定値とエンコーダカウンタを比較する
STOP X_A C_MORE /* トリガ条件設定(パルス停止ではない)
CLRPOS -1 /* カウンタクリア
COMP=0
OFF 0
DO
COMP=COMP+1000
dummy=0
RANGE X_A COMP dummy /* エリア設定
a=RR3(X_A) /* 割り込みフラグリセット
IF SW(0)==0 THEN
INTA_ON 0 (0,X_A) /* X(-1)>=COMP で ON 0
WAIT HSW(0)==1
ELSE
INTA_OFF 0 (0,X_A) /* X(-1)>=COMP で OFF 0
WAIT HSW(0)==0
END_IF
PRINT X(-1) SW(0)
LOOP
END

```

・ 実行結果

ロータリエンコーダをCW方向に回すと1000カウント毎に出力をon/offします

```

1000 1
2000 0
3000 1
4001 0 /* この場合インタプリタのディレイでトリガタイミングと表示までに位置がずれることがある


```

■サンプル5 Up/Downカウンタの読み込み

INSETでUP_DWN(予約定数)を指定するとUp/Downカウンタになります(Up/Downカウンタは常に1倍)。

読み込む方法はサンプル1と同じです。

次のサンプルプログラムは右図のように同一のMPG-2314のパルス出力とエンコーダ入力を接続して作成しました。



```

PG 0
ACCEL X_A 10000
INSET UP_DWN /* Up/Downカウンタモード
CLRPOS -1 /* カンタクリア
CLRPOS /* パルス現在値クリア
P=10000
DO
P=P*-1
MOVS X_A P /* パルス発生
WAIT RR(X_A)==0 /* パルス発生終了待ち
PRINT X(-1) /* Xカウンタ現在値表示
TIME 100
LOOP

```

・ 実行結果

パルス発生終了毎にXカウンタの現在値を表示します。

```

-10000
10000
-10000
10000

```

■サンプル6 リングカウンタ

RANGE コマンドで VRING を指定するとリングカウンタになります。
1000p/r のロータリエンコーダではジャスト1周で0に戻ります。

```
PG 0
RANGE VRING|X_A 999          /* MCX314 リングカウンタ設定 0~999
INSET PHASE1
CLRPOS -1
DO
  NOWC=X(-1)
  PRINT NOWC
  WAIT NOWC<>X(-1)
LOOP
```

- 実行結果 (0 前後で往復)

```
0
1
2
3
2
1
0
999
998
999
0
1
2
1
0
999
998
997
998
999
0
1
2
3
4
5
6
```

次の例は*TASK1 で X 軸からパルス発生して、*TASK2 でそれをカウントしています。(結線はサンプル5 参照)
注意点は

- 1)RANGE VRING を設定すると出力カウンタもリングカウンタになります。出力側の絶対座標管理が必要な場合はパルス出力軸とカウンタ入力軸を違えます。
- 2)RANGE の前に ACCEL を行います。

```
*TASK1
PG 0                          /* PG 宣言
ACCEL X_A 100 100 100        /* ACCEL は RANGE の前に実行

INSET X_A UP_DWN            /* カウンタ種別
RANGE VRING|X_A 359         /* リングカウンタ
CLRPOS X_A -1               /* カウンタクリア

CLRPOS X_A                  /* 出力座標クリア
RMVC X_A CW                 /* 連続パルス出力

END

*TASK2
PG 0                          /* PG 宣言
DO
  PR X(0) X(-1)             /* 出力座標, カウンタ値
  TIME 5
LOOP
```

- 実行結果

```
RUN

# 1 0
1 1
2 1
```

2 2
3 2
(略)
358 357
358 358
359 358
359 359
0 359 ← X(0)は0に戻る
1 0 ← X(-1)も0に戻る
1 1
2 1
(略)
358 357
359 358
359 359
0 359
0 0
1 0
(略)

■サンプル7 リアルタイムトリガ出力


サンプル4と同じように指定したポイントでトリガ信号を出します。
この方法はMPG-2314に搭載しているPG-ICの機能を使用しているので精度の良い信号が出せます。

次のプログラムはMPG-2314 J4 コネクタ O0 出力(19 番ピン)から、最初は 1000 パルスで、次から 200 パルス毎にトリガ信号を出します。

[注意] 信号幅(Low になっている時間)はタスクの数や状況によって数 10 μ ~1msec 程度の変動します。精度が要求される場合はシングルタスクで用いてください。

```

CLR_OUTP 1
PG 0
ACCEL X_A|OUTSL 10000 /* カウンタ値==設定値で J4 19 番ピンが ON する(この場合の 10000 は意味無し)
INSET X_A CMP_CNT|PHASE1 /* カウンタ入力の設定
P_DET=1000 /* 最初のトリガポイントのパルス数
INTVL=200 /* 2 回目以降の間隔
RANGE X_A P_DET 0 /* 最初のトリガポイントの設定
CLRPOS -1 /* カウンタクリア
DO
  dmy=CMP_C(0, X_A) /* カウンタの比較。この場合通常出力 0 もオンします。
  P_DET=P_DET+INTVL /* 次のトリガポイントを算出
  RANGE X_A P_DET 0 /* 次のトリガポイントを設定。ここで 00 が OFF される
LOOP
  
```



トリガ出力に通常出力ポートを使用することもできます。しかし、上のプログラムでは最初の O0 がオンされるのと同時に出力 0 もオンされますがオフにはなりません。

次のプログラムは TIME 後に出力 0 をオフにしています。ただし、TIME 値がトリガの間隔より長くなると動作しません。

[注意] 出力 0 がオンになるタイミング、信号幅はタスクの数や状況によって数 10 μ ~1msec 程度の遅延・変動します。精度が要求される制御には不向きです。

```


CLR_OUTP 1
PG 0
ACCEL X_A|OUTSL 10000
INSET X_A CMP_CNT|PHASE1
P_DET=1000
INTVL=200
RANGE X_A P_DET 0
CLRPOS -1
DO
  dmy=CMP_C(0, X_A) /* 論理/実位置カウンタ==COMP レジスタで出力 0 オン。
  P_DET=P_DET+INTVL
  RANGE X_A P_DET 0
  PR X(-1) /* (注意) 出力間隔が超短時間の場合 PRINT 文があると追従できないことがある。
  TIME 10 /* (注意) TIME >= トリガ間隔 は NG
  OFF 0 /* 出力 0 オフ
LOOP
  
```

- 実行結果
 - 1000 ←最初のトリガポイント
 - 1200 ←2 回目のトリガポイント
 - 1400
 - 1600
 - 1800
 - 2000

■ サンプル 8 ダブルトリガ出力

・ 機器構成

このサンプルプログラムは U 軸 CW パルス出力を X 軸カウンタ入力に接続して実行しています。



・ 概要

MPG-2314 のカウンタ入力に同調してトリガ信号を出力します。更なるその信号から任意の遅延時間後にもう一つのトリガ信号を出力します。それを一定間隔で繰り返します。

・ 動作原理

この機能は MPG-2314 に搭載している MCX314 の「カウンタ比較」と「同期動作」を応用しています。

「カウンタ比較」は論理・実位置カウンタと比較レジスタの値を比べて出力・レジスタを操作します。

「同期動作」は” IC 内の各軸内、軸間、および IC 外のデバイスとの間において、ドライブ開始・停止などの動作 (Action) を連携して行う機能のことです” (MCX314 取扱説明書より)。

このサンプルのトリガ発生は X、Z 軸のカウンタ比較と同期動作を組み合わせで実現しています。

・ 使用ボード

MPC-2000(SH7030) BL/I 1.12_87 2011/12/13、 MPG-2314 CEP-128D

・ MPC プログラム

コマンドパラメータの説明

ACCEL コマンド

OUTSL : カウンタ値 \geq 比較レジスタ値で MPG-2314 の O0~O3 出力をオンする

INSET コマンド

CMP_CNT : 比較レジスタとカウンタ入力の比較する

UP_DWN : Up Down カウンタ入力(2相エンコーダを接続する場合は PHASE1)

RANGE コマンド

比較レジスタの設定(この場合はトリガを出力する位置を指定)

SYNC コマンド

同期動作設定。同期させる他軸の選択と起動要因の設定。他軸の動作(Action)指定。


CLRPOS コマンド

実位置とカウンタ入力のクリア

CMP_C 関数

カウンタ値と比較レジスタ値の比較結果。サンプルではカウンタ値 \geq 比較レジスタ値で 0 以外になる。

動作イメージ



/* このサンプルは U軸パルス->X Counter に接続し、そのパルス数でトリガを出す
/* 2相エンコーダを接続するときは ※1行 を削除、※2行を有効にする。

PG 0

ACCEL U_A 10000 100 10000 /* U軸パルス発生設定 ※1

/* 03(J4 22番ピン)トリガ遅延時間=1/z_freq*DLY。z_freq=1000000、DLY=10なら 1μ*10
z_freq=1000000

ACCEL Z_A|OUTSL z_freq 10000 z_freq /* 03トリガ Enable
ACCEL X_A|OUTSL 3000 /* 00トリガ Enable (3000はdummy)

INSET X_A CMP_CNT|UP_DWN /* カウンタ入力の設定: Up/Downカウンタ ※1
' INSET X_A CMP_CNT|PHASE1 /* カウンタ入力の設定: 2相エンコーダ ※2

SYNC X_A &H00004001 0 /* 同期動作設定。同期動作させる他軸: Z軸、起動要因: P≥C+
SYNC Z_A 0 4 /* 同期動作設定。+方向連続パルスドライブを起動する

CLRPOS /* 現在位置クリア
CLRPOS -1 /* カウンタクリア

P_DET=200 /* 最初の00トリガポイントのパルス数
DLY=10 /* 03トリガ遅延時間
OFS=0
RANGE Z_A DLY 0 /* 03トリガ遅延時間設定

RMVC U_A 1 /* U軸パルス発生-> X軸カウンタ ※1


DO

RANGE X_A P_DET+OFS 0 /* 00トリガポイント設定 & 00トリガ off
OFS=OFS+100 /* 2回目以降の間隔


WAIT CMP_C(Z_A) !=0 /* 03トリガ発生確認(トリガと同時にZ軸からパルス出力が開始される)
STOP Z_A STP_I /* パルス発生停止
WAIT RR(Z_A) ==0 /* パルス発生停止待ち
' TIME 1 /* パルス幅(注意: TIME値に加えて約1msecの変動あり)
CLRPOS Z_A /* Z軸現在位置クリア & 03トリガ off

WAIT CMP_C(X_A) !=0
LOOP


・実行結果



time range 10µsec
トリガ発生部分



time range 200µsec
1ショット
[注意] タスク数、動作状態によってパルス幅が変動するので注意してください。



time range 10msec
連続状況

■Appendix DC24V I/F 機器の接続

